arXiv:gr-qc/0403031 v1 7 Mar 2004

Calculating symmetries in Newman-Tamburino metrics

John D. Steele School of Mathematics, University of New South Wales, Sydney, New South Wales 2052, Australia. email: j.steele@unsw.edu.au

Abstract: In this paper I show that the Newman-Tamburino spherical metrics always admit a Killing vector, correcting a claim by Collinson and French, (1967 J. Math. Phys. 8 701) and also admit a homothety. A similar calculation is given for the limit of the Newman-Tamburino cylindrical metric.

1 Introduction

The Newman-Tamburino metrics are those vacuum solutions of the Einstein equations admitting hypersurface orthogonal geodesic rays with non-vanishing shear and divergence. In the Newman-Penrose formalism this implies that Ψ_0 = κ = 0, that ρ is real and nonzero and σ ≠ 0. In [1] Newman and Tamburino explicitly gave all such metrics and showed that they fall into two classes: the spherical, with ρ^2 ≠ σσ and the cylindical with ρ^2 = σσ. In [2] Collinson and French claimed to have shown that the former metrics admit at most one Killing vector, and that happens only in a particular subcase. In fact, the spherical Newman-Tamburino metrics always admit a Killing vector and also always admit a homothety. This preprint is intended to show the full calculations and results when the homothetic equations of [3] are integrated for the Newman-Tamburino spherical metrics. The bulk of sections 3 and 4 come from Maple 9 worksheets, exported to TEX and suitably tidied up for better readability.

Throughout I use the spin coefficient notation of [4]. For example I use κ^′, ρ^′, σ^′ and τ^′ in place of the more traditional −ν, −μ, −π and −λ.

2 Results

The contravariant form of the Newman-Tamburino spherical metric [1] (see also [5], equation (26.21)) is

g^{22} = −

2r^2(ζζ)^{1/2} R^2 + 2rL A + 2r^3A(ζ^2 + ζ^2) R^4 − 4r^2A^2(ζζ)^{3/2} R^4

g^{23} = 4A^2(ζζ)^{3/2}x [ L 2a^3 − r − 2a 2a^2R^2 − r − a R^4 ]

g^{24} = 4A^2(ζζ)^{3/2}y [ L 2a^3 − r + 2a 2a^2R^2 − r + a R^4 ]

g^{33} = −

2(ζζ)^{3/2} (r + a)^2 g^{44} = −

2(ζζ)^{3/2} (r − a)^2 g^{12} = 1.

Here our coordinates are x^1 = u, x^2 = r, x^3 + ix^4 = x + iy = ζ and A(u) = bu + c, L = 1 2 log ( r + a r − a ) a = A(ζζ)^{1/2} R^2 = r^2 − a^2.

Here b and c are real constants.

The Collinson and French result (also quoted in [5]) is that there is a Killing vector only in the case where A is constant — in this situation the Killing vector is the obvious ∂_u. However, if b ≠ 0 we can set c = 0 by a coordinate change and then the vector

K^a = −u∂_u + r∂_r + 2x∂_x + 2y∂_y .

is a Killing vector, as will be shown in section 3. This can be checked directly: consider the flow of K^a. This scales the coordinates by u → λ^{−1}u, r → λr, ζ → λ^2ζ for real parameter λ > 0. Under this scaling it is easy to check that all the contravariant components given above are homogeneous in λ (when A = bu), and all of the correct degree to make the flow isometric. For example, the g^{22} component is homogeneous of degree 2, and so the metric term g^{22} ∂ ∂r ⊗ ∂ ∂r is unchanged under the flow.

Also the vector H = r∂_r + x∂_x + y∂_y is a homothety, whatever A is (see section 3). Alternatively, the flow of H is u → u, r → λr, ζ → λζ, and we again find that all the contravariant components given above are homogeneous in λ, and all of the correct degree to make the flow homothetic. For example, the g^{22} component is homogeneous of degree 1, and so the metric term g^{22} ∂ ∂r ⊗ ∂ ∂r scales by λ^{−1}: the same scaling applies to all the metric terms.

Newman and Tamburino [1] also give the following metric, which arises as a limit of the cylindrical case (see also [5] (26.23) for corrections to the du^2 coefficient): ds^2 = 2 du dr − x^{−2} [ b + log(r^2x^4) ] du^2 + 4 r x du dx − r^2dx^2 − x^2dy^2, with the same coordinates as used in the spherical case. The Killing vectors here are obvious (∂_u and ∂_y) and as we shall see there is also a homothetic vector (see section 4)

H_2 = 2r∂_r − x∂_x + 2y∂_y.

One can use the flow of H_2 to check it is a homothety as well.

3 The Calculations (spherical case)

The basic information is taken from Collinson and French [2], and Newman and Tamburino [1]. See those papers for those spin coefficients that are not actually calculated here. I have checked in a separate calculation that their results are correct as quoted. I use as coordinates u, r, ζ = x + iy.

Collinson and French [2] wrote the conformal Killing equations in Newman-Penrose form and used that in their work, although there are a few minor typos in their paper. Here, I will use the formalism of [3], which generalised the ideas of [6] into a form suitable for this task. I will use to the notation of [3] for the components of the homothety

ξ_a = ξ_nℓ_a + ξ_ℓn_a − ξ_mm_a − ξ_mm_a,

and its bivector, F_{ab}, with anti-self dual

−F_{ab} = 2ϕ_{00} ℓ_{[a}m_{b]} + 2ϕ_{01} (ℓ_{[a}n_{b]} − m_{[a}m_{b]}) − 2ϕ_{11} n_{[a}m_{b]}.

The tetrad is a standard tetrad (see [1]), based around the Debever-Penrose vector ℓ^a = ∂_r, see [1] and [2] for further detail. Since the tetrad is normalised, for the Penrose-Rindler spin coefficients used in [3] we have γ^′ = −ϵ, β^′ = −α etc.

In the Maple I use use z for ζ and w for ζ; Hl for ξ_ℓ etc. I typically add a b for a complex conjugate (Hmb is ξ_m) and a 1 for a dash (rho1 is ρ^′).

Firstly, define the terms a, a0 (the latter is α_0 in [2]).

> a:=A(u)*z^(1/2)*w^(1/2):
> a0:=3/4*w^(3/4)*z^(-1/4):
> a0b:=3/4*z^(3/4)*w^(-1/4):

Rather than use the explicit definitions for L and R in [2] and [1], I will leave them as “unknown” functions and define a routine later that will substitute for their derivatives. I will also use Q(u, r, z, w) in place of 1/R^2 to make things more transparent. I define what these functions actually are so we can substitute for them more easily when that become useful. I also define dummy symbols to use in place of the full functional dependence of L and Q for ease of readability. I have also suppressed the functional dependence in the Maple output, replacing Q(u, r, z, w) with Q(x^a) for example.

> LL:=L(u,r,z,w):Lis:=1/2*log((r+a)/(r-a));

Lis := 1 2 ln ( r + A (u) √ z √ w r − A (u) √ z √ w )

> QQ:=Q(u,r,z,w):Qis:=1/(r^2-a^2);

Qis := ( r^2 − (A (u))^2 zw )_{−1} Now we define the routine to simplify derivatives and products and also add a line to collect terms.

> diffsbs:=proc(XX)
> subs(diff(L(u,r,z,w),r)=-a*Q(u,r,z,w),XX):
> subs(diff(L(u,r,z,w),u)=r*Q(u,r,z,w)*diff(a,u),%);
> subs(diff(L(u,r,z,w),w)=r*Q(u,r,z,w)*diff(a,w),%):
> subs(diff(L(u,r,z,w),z)=r*Q(u,r,z,w)*diff(a,z),%):
> subs(diff(Q(u,r,z,w),r)=-2*r*Q(u,r,z,w)^2,%):
> subs(diff(Q(u,r,z,w),z)=2*Q(u,r,z,w)^2*a*diff(a,z),%):
> subs(diff(Q(u,r,z,w),u)=2*Q(u,r,z,w)^2*a*diff(a,u),%):
> subs(diff(Q(u,r,z,w),w)=2*Q(u,r,z,w)^2*a*diff(a,w),%):
> student[powsubs](r^2=a^2+1/Q(u,r,z,w),expand(%));
> collect(%,[L(u,r,z,w),Q(u,r,z,w),r,Hl(u),psi,z,w]);
> end proc:

The terms S and Sb are ψ^1_0 and its conjugate in [2].

> S:=2*A(u)^2*z^(3/4)*w^(3/4)*z;Sb:=2*A(u)^2*z^(3/4)*w^(3/4)*w: S := 2 A (u)^2 z^{7/4}w^{3/4} And the curvature component Ψ_1 is given in [1]. > Psi1:=S*QQ^2;

Ψ1 := 2 A (u)^2 z^{7/4}w^{3/4}Q (x^a)^2 Now from [2] we have κ = ϵ = τ^′ = Ψ_0 = 0, and ρ and σ real — these can also be easily checked by Maple. So by [3](6a) Dξ_ℓ = 0. Using τ = α + β ([2]), [3](6c) becomes δξ_ℓ = τ ξ_ℓ − ρξ_m − σξ_m − ϕ_{11}. We next use equation [3](11), since ℓ^a is a Debever-Penrose vector. Unfortunately, [3](11) contains an error — the right hand side is the complex conjugate of what it ought to be. With this correction, we have

ϕ_{11} = −τ ξ_ℓ + ρξ_m + σξ_m.

Hence δξ_ℓ = 0 and ξ_ℓ = ξ_ℓ(u), as found in [2].

Equation (10a) of [3] is Dϕ_{11} = −ξ_ℓΨ_1, and so integrates to give the r dependence of ϕ_{11}, here called p11. We ignore the factor independent of r when integrating:

> int(-Qis^2,r);

− r 2A (u)^2 zw (r^2 − A(u)^2zw) + 1 2 arctanh ( r A (u) √ zw ) A (u)^{−3} z^{−1}w^{−1} 1 √ zw The arctanh term here is just L and we get

> p11:=S*Hl(u)/2/a^2*r*QQ-S*Hl(u)/2/a^3*LL+p110(u,z,w);

p11 := z^{3/4}Hl (u) rQ (x^a) w^{1/4} − z^{1/4}Hl (u) L (x^a) A (u) w^{3/4} + p110 (u, z, w)

Here p110(u,z,w) is the integration constant. Now the spin coefficients — see [2].

> alpha:=expand(simplify(S*LL*r/2/a^2*QQ+QQ*(r*a0+a*a0b)-S*QQ/a/2,radical)):
> beta:=-S*LL*QQ/2/a-QQ*(r*a0b+a*a0):

Now a routine to take conjugates nicely, as we need conjugates to define τ .

> conj:=proc(XX)
> subs(z=w1,w=z1,XX):
> subs(w1=w,z1=z,%):subs(I=-I,%);
> subs(L(u,r,w,z)=L(u,r,z,w),Q(u,r,w,z)=Q(u,r,z,w),%);
> subs(p110(u,w,z)=p110b(u,z,w),p110b(u,w,z)=p110(u,z,w),%);
> end proc:
> conj(alpha)+beta:
> tau:=collect(%,[L(u,r,z,w),Q(u,r,z,w)]);

τ := ( −A (u) z^{5/4}w^{1/4} + rw^{3/4}z^{−1/4} ) Q (x^a) L (x^a) − A (u) w^{5/4}z^{1/4}Q (x^a)

> rho:=-r*Q(u,r,z,w):
> sigma:=a*Q(u,r,z,w):
> sigma1:=r*LL^2*S^2/4/a^4*QQ+r*LL*S*Sb/2/a^4*QQ-r*diff(a,u)*QQ-S*Sb/2/a^3*QQ:
> unprotect(gamma);
> gamma:=-r*LL^2*S*Sb/4/a^4*QQ+S^2*LL^2/4/a^3*QQ+a*r*LL*(S*a0b-Sb*a0)/2/a^3*QQ
> + S*Sb*QQ*LL/2/a^3 - S*Sb/4/a^3*(LL/2/a^2-r/2/a*QQ) +
> (S^2/2/a-a*(S*a0b-Sb*a0))/2/a^2*QQ:

We need derivative operators δ and δ^′ to find ρ^′. Firstly, the components of m^a come from [2] and [1].

> om0:=-A(u)*z^(1/4)*w^(5/4):
> omega:=-Sb*LL/2/a^2+(r*om0-a*conj(om0))*QQ;

ω := −w^{3/4}z^{−1/4}L (x^a) + ( −rA (u) z^{1/4}w^{5/4} + A (u)^2 z^{7/4}w^{3/4} ) Q (x^a)

> omega1:=conj(omega):
> P:=z^(3/4)*w^(3/4):
> del:=XX->omega*diff(XX,r)+2*r*P*QQ*diff(XX,w)-2*a*P*QQ*diff(XX,z):
> del1:=XX->omega1*diff(XX,r)+2*r*P*QQ*diff(XX,z)-2*a*P*QQ*diff(XX,w):

To calculate ρ^′ we use [4] (4.11.12 e^′).

> expand((diff(sigma1,r)-rho*sigma1)/sigma):
> rho1:=diffsbs(%);

ρ1 := −A (u) z^2L (x^a)^2 Q (x^a) − 2 ( A (u) zw + z^{3/2}r √ w ) Q (x^a) L(x^a) + A (u) zw d A du Q (x^a)

We check some curvature equations next before we go on.

> diffsbs(diff(tau,r) - rho*tau-sigma*conj(tau)-Psi1); #[4]4.11.12c 0

> diffsbs(diff(alpha,r)-rho*alpha-sigma*beta); # [4] 4.11.12h & i’; 0 > diffsbs(diff(beta,r)-sigma*alpha-beta*rho-Psi1); #[4]4.11.12h’& i; 0 > Psi2:=-diffsbs(diff(rho1,r)-rho1*rho-sigma*sgma1) ; # [4] 4.11.12 f’ Ψ2 := −4A (u)^2 z^{5/2}L (x^a) Q (x^a)^2 √ w − ( 2 A (u) z^2r + 4A (u)^2 z^{3/2}w^{3/2} ) Q (x^a)^2 This expression for Ψ_2 agrees with [1].

> diffsbs(diff(gamma,r) -beta*conj(tau)-alpha*tau-Psi2); # [4]4.11.12k 0

> diffsbs(del(rho)-del1(sigma)-rho*(conj(alpha)+beta)+
> sigma*(3*alpha-conj(beta) ) + Psi1); # [4] 4.11.12 d

0

> diffsbs( del1(beta)-del(alpha)-rho*rho1+sigma*sigma1+alpha*conj(alpha)+
> beta*conj(beta)-2*alpha*beta-Psi2 ); # [4] 4.11.12 l

0 Integrating [3] (6g) and using [3](11) (corrected, see above): > Hm1:=-r*p110(u,z,w) + Hl(u)*S/2/a^3*r*LL+Hm0;

Hm1 := −rp110 (u, z, w) +

Hl (u) z^{1/4}rL (x^a) A (u) w^{3/4} + Hm0

> Hmb1:=-r*p110b(u,z,w) + Hl(u)*Sb/2/a^3*r*LL+Hmb0;

Hmb1 := −rp110b (u, z, w) +

Hl (u) w^{1/4}rL (x^a) A (u) z^{3/4} + Hmb0

By [3] (11) the following ought to be zero.

> diffsbs(p11+tau*Hl(u)-rho*Hm1-sigma*Hmb1):
> collect(%/QQ,r);

( Hm0 + z^{3/4}Hl (u) w^{1/4} + A (u) √ z √ w p110b (u, z, w) ) r − A (u) √ z √ wHmb0 − (A (u))^2 zw p110 (u, z, w) − Hl (u) A (u) w^{5/4}z^{1/4}

> expand(solve(coeff(%,r,1),Hm0)),expand(solve(coeff(%,r,0),Hmb0));

− z^{3/4}Hl (u) w^{1/4} − A (u) √ z √ wp110b (u, z, w) , − √ z √ wA (u) p110 (u, z, w) − w^{3/4}Hl (u) z^{1/4} So we get > Hm:=-r*p110(u,z,w)-a*p110b(u,z,w)+Hl(u)*expand(S/2/a^3*(r*LL-a)):

> Hmb:=conj(Hm):

These agree with the components in [2] (their V_3 and V_4). Now we use [3] (10b) to get ϕ_{01}.

> diffsbs(Psi1*Hm/2/sigma-beta*p11/sigma+del(p11)/2/sigma):
> p01:=collect(%,[Hl(u),Q(u,r,z,w),L(u,r,z,w),r]);

p01 := (( 1 2 √ z √ wr + ( 2 z^{3/2}r √ w + A (u) zw ) L (x^a) ) Q (x^a) − z (L (x^a))^2 A (u) w − L (x^a) 2A (u) ) Hl (u) + z^{3/4}L (x^a) p110 (u, z, w) w^{1/4} + 3w^{3/4}p110 (u, z, w) 4z^{1/4} − ( A (u) z^{5/4}w^{1/4}r p110 (u, z, w) + A (u)^2 z^{7/4}w^{3/4}p110b (u, z, w) ) Q (x^a) + ( 3p110 (u, z, w) 4A (u) w^{3/4} + w^{1/4 ∂}_{∂w} p110 (u, z, w) A (u) ) z^{1/4}r − z^{3/4}w^{3/4} ∂ ∂z p110 (u, z, w) Now [3](10c) and (8a) will give us information on the w (that is, ζ) dependence of ϕ^0_{11}.

> diffsbs(Psi1*Hmb-Psi2*Hl(u)+2*rho*p01+2*alpha*p11-del1(p11)); # [3]10c ( 3p110 (u, z, w) 2A (u) w^{3/4} − 2 w^{1/4} A (u) ∂ ∂w p110 (u, z, w) ) z^{1/4} > dsolve(%=0,p110(u,z,w));

p110 (u, z, w) = F1 (u, z) w^{3/4} > diffsbs(diff(p01,r)+del1(p11)-2*rho*p01-2*alpha*p11); # [3] (8a) ( 9p110 (u, z, w) 4A (u) w^{3/4} + 3 w^{1/4} A (u) ∂ ∂w p110 (u, z, w) ) z^{1/4} > dsolve(%=0,p110(u,z,w));

p110 (u, z, w) = F1 (u, z) w^{3/4} Both giving the same result. Now we turn to ξ_n and [3](6i), which we solve for σξ_ℓ.

> rhs6i:=diffsbs((-del(Hm)-conj(sigma1)*Hl(u)-Hm*(conj(alpha)-beta))): The imaginary part ought to be zero as ξ_n is real, so using results from [3] (10c) and (8a), we find the imaginary part divide out a common nor-zero factor and call what’s left X.

> Imrhs6i:=%-conj(%):
> subs(p110(u,z,w)=F(u,z)/w^(3/4),p110b(u,z,w)=Fb(u,w)/z^(3/4),%):
> X:=expand(%/r/QQ/sqrt(z)/sqrt(w));

X := −4 z^{3/4}A (u) ∂ ∂z F (u, z)+4 w^{3/4}A (u) ∂ ∂w Fb (u, w)+3 A (u) Fb (u, w) w^{1/4} −3 A (u) F (u, z) z^{1/4} Assuming F is differentiable in z we can split this

> subs(Fb=0,X):%;

−4 z^{3/4}A (u) ∂ ∂z F (u, z) − 3 A (u) F (u, z) z^{1/4} This is a (real) function of u and w. We choose the shape of the separation function to simplify the solution to the differential equation slightly.

> dsolve(%=-4*A(u)*G(u),F(u,z)); F (u, z) =^{G(u)z+ F1 (u)}_{z3/4} Check this out:

> subs(F(u,z)=G(u)*z^(1/4)+H(u)/z^(3/4),Fb(u,w)=G(u)*w^(1/4)+Hb(u)/w^(3/4),X):
> expand(%);

0 So we define a simplification routine for ϕ^0_{11}.

> P110sbs1:=proc(XX)
> subs(p110(u,z,w)=F(u,z)/w^(3/4),p110b(u,z,w)=Fb(u,w)/z^(3/4),XX);
> subs(F(u,z)=G(u)*z^(1/4)+H(u)/z^(3/4),Fb(u,w)=G(u)*w^(1/4)+Hb(u)/w^(3/4),%);
> expand(%);
> end proc:

And check it works > P110sbs1(Imrhs6i); 0 Turning to [3](6b) next,

> eqn6b:=Hl(u)*(gamma+conj(gamma))-conj(tau)*Hm-tau*conj(Hm)-p01-conj(p01)+psi:
> P110sbs1(diffsbs(%));

ψ − G (u) − 3 H (u) 2z − 3 Hb (u) 2w This ought to be ξ̇_ℓ, a function of u only, so H = 0 and we define a new simplification routine and test it out:

> P110sbs2:=proc(XX);
> expand( subs( p110(u,z,w)= (psi-diff(Hl(u),u))*z^(1/4)/w^(3/4) ,
> p110b(u,z,w)=(psi-diff(Hl(u),u))*w^(1/4)/z^(3/4),XX));
> collect(%,[L(u,r,z,w),Q(u,r,z,w),r,Hl(u),z,w]);end proc:
> P110sbs2(diffsbs(eqn6b));

d du Hl (u) This is as it should be. Now we can define ξ_n.

> P110sbs2(diffsbs(rhs6i+conj(rhs6i))/sigma/2):
> Hn:=collect(%,[L(u,r,z,w),Q(u,r,z,w),Hl(u),diff(Hl(u),u),r,psi]):

We check this against the [2] version, called V_2 there. It is clear from the shape of ξ_m (= V_3 of [2]) that a_0 in [2] is my ϕ^0_{11}.

> ay0:=p110(u,z,w):ay0b:=p110b(u,z,w):
> V2:=r*LL^2*(-S^2-Sb^2)*Hl(u)/4/a^5 - LL^2*S*Sb*Hl(u)/4/a^4 -
> LL*(ay0b*S+ay0*Sb)/2/a + r*(2*a*Hl(u)*diff(a,u)-ay0*Sb-ay0b*S)/2/a^2 +
> (-2*a^3*ay0*S-2*a^3*ay0b*Sb+Hl(u)*S*Sb)/4/a^4+r*LL*(-2*a^3*ay0*S-
> 2*a^3*ay0b*Sb- Hl(u)*S*Sb)/4/a^5 + 1/QQ*(-3*ay0*S-3*ay0b*Sb-
> 4*a^2*2*P*(diff(p110(u,z,w),w) +diff(p110b(u,z,w),z) ) )/8/a^3:
> expand(Hn-P110sbs2(V2)):
> simplify(subs(psi=0,diff(Hl(u),u)=0,%));

0 So our ξ_n agrees with [2] in the case of their Killing vector (ψ = 0 and ξ_ℓ constant).

However, if ξ_ℓ is not constant, the terms differ:

> simplify(subs(psi=0,%%));

− 2 z^{3/2}w^{3/2} dHl (u) du (( w^3z + z^3w + 2z^2w^2L (x^a) ) A (u) + L (x^a) r(z^{5/2} √ w + w^{5/2} √ z) + 2rz^{3/2}w^{3/2} )

Next, we put our ξ_n into [3](6d).

> eqn6d:=diffsbs(diff(Hn,r)-p01-conj(p01)-psi):
> P110sbs2(%);

Hl (u)^d_{du} A (u) A (u) − d du Hl (u)

> dsolve(%,Hl(u)); Hl (u) = C1 A (u) So next a routine to replace ξ_ℓ(u) with a multiple of A(u), and also to kill off the second derivative of A(u).

> Hlsbs:=proc(XX)
> subs(Hl(u)=C*A(u),XX);subs(diff(A(u),u,u)=0,%);subs(diff(Hl(u),u,u)=0,%);%;
> end proc:

We now try [3](6j).

> del1(Hm)+conj(rho1)*Hl(u)+rho*Hn+(conj(beta)-alpha)*Hm-p01+conj(p01)+psi:
> Hlsbs(P110sbs2(diffsbs(%)));

0

So that is satisfied. Now for ϕ_{00}, which we get from the conjugate of [3](6f).

> eqn6f:=-del1(Hn)-(conj(beta)+alpha)*Hn-conj(rho1)*Hmb-sigma1*Hm:
> p00:=diffsbs(P110sbs2(diffsbs(%))):

I’ve supressed this component as it’s very long, but we check the result with [3] (8d).

> del1(p01)+diff(p00,r)-rho*p00-sigma1*p11:
> P110sbs2(diffsbs(%));

0 To go any further we need to get the components of n^a = (1, U, X^3, X^4) and to define D^′. Taking the metric terms from [1] and [2]:

> gup22:=-2*r^2*sqrt(w)*sqrt(z)*QQ + 2*r*LL/A(u) + QQ^2*(2*r^3*A(u)*(w^2+z^2)
- 4*r^2*A(u)^2*w^(3/2)*z^(3/2)):
> gup22+2*omega*conj(omega):
> U:=diffsbs(%/2);

U : = √ z √ wL (x^a)^2 + (( w^2 + z^2 ) A (u) r − 2 A (u)^2 z^{3/2}w^{3/2} ) Q (x^a) − √ z √ w + ( ( −A (u)^2 z^{5/2} √ w + 2 zwrA (u) − A (u)^2 √ zw^{5/2} ) Q (x^a) + r A (u) ) L (x^a)

> gup33:=-2*z^(3/2)*w^(3/2)/(r+a)^2:gup44:=-2*z^(3/2)*w^(3/2)/(r-a)^2: These next two terms are the components of m^a. > xi3:=P*(r-a)*QQ;xi4:=I*P*(r+a)*QQ; xi3 := z^{3/4}w^{3/4} (r − A (u) √ z √ w) Q (x^a) xi4 := iz^{3/4}w^{3/4} (r + A (u) √ z √ w) Q (x^a) > xi3*conj(xi4)+xi4*conj(xi3); # checking 0 > simplify(( subs( Q(u,r,z,w)=Qis,xi3*conj(xi3)*2+gup33) )); 0 > simplify(subs(Q(u,r,z,w)=Qis,xi4*conj(xi4)*2+gup44)); 0

> gup23:=4*A(u)^2*z^(3/2)*w^(3/2)*(z+w)/2*(LL/2/a^3-(r-2*a)*QQ/2/a^2
> -(r-a)*QQ^2):
> gup24:=4*A(u)^2*z^(3/2)*w^(3/2)*(z-w)/2/I*(LL/2/a^3-(r+2*a)*QQ/2/a^2
> -(r+a)*QQ^2):
> gup23+omega*conj(xi3)+conj(omega)*xi3:
> X3:=diffsbs(%);

X3 := ( (( −z^{3/2} √ w − √ zw^{3/2} ) r + A (u) zw^2 + A (u) z^2w ) Q (x^a) + w A (u) + z A (u) ) L (x^a) + (( −z^{3/2} √ w − √ zw^{3/2} ) r + A (u) zw^2 + A (u) z^2w ) Q (x^a)

> gup24+omega*conj(xi4)+omega1*xi4:
> X4:=diffsbs(factor(%));

X4 := i ( (( −z^{3/2} √ w + √ zw^{3/2} ) r + zw^2A (u) − z^2wA (u) ) Q (x^a) + w A (u) − z A (u) ) L (x^a) + (( −i √ zw^{3/2} + iz^{3/2} √ w ) r + iz^2wA (u) − izw^2A (u) ) Q (x^a)

As a double check we firstly define the (contravariant) tetrad and then check against the metric terms.

> ell:=<0,1,0,0>:en:=<1,U,X3,X4>:
> em:=<0,omega,xi3,xi4>:emb:=map(conj,em):
> ell.Transpose(en)-em.Transpose(emb):
> %+Transpose(%):
> g:=map(diffsbs,%):
> diffsbs(g[2,2]-gup22);

0 > diffsbs(g[2,3]-gup23); 0 > diffsbs(g[2,4]-gup24); 0 > diffsbs(simplify(g[3,3]-gup33)); 0 > diffsbs(simplify(g[4,4]-gup44)); 0 For a second check we apply two of the commutators [4] (4.11.11) to r and check what we get.

> diff(U,r)+gamma+conj(gamma)-tau*conj(omega)-conj(tau)*omega:
> diffsbs(%);

0

> diff(X3,r)-tau*conj(xi3)-conj(tau)*xi3:
> diffsbs(%);

0

Since all this checks out we go ahead and define D^′.

> D1:=proc(XX)
> diff(XX,u)+diff(XX,r)*U+(X3+I*X4)*diff(XX,z)+(X3-I*X4)*diff(XX,w);
> P110sbs2(diffsbs(%));
> end proc:

We make use of D^′ firstly to find the last spin coefficient, κ^′, using [4](4.11.12g).

> D1(beta)-del(gamma)-tau*rho1-alpha*conj(sigma1)-beta*(rho1+gamma-conj(gamma))
> +gamma*(tau-beta-conj(alpha)): # should be -kappa’*sigma
> diffsbs(%):
> kappa1:=-diffsbs(P110sbs2(%/sigma));

κ1 := − ( z^{5/4}w^{1/4}r − z^{11/4}A (u) w^{1/4} ) Q (x^a) L (x^a)^3 − (( −2 z^{9/4} w^{3/4} + 2 z^{1/4}w^{5/4} ) r − 3 z^{7/4}w^{3/4}A (u) ) Q (x^a) L (x^a)^2 + (( d A du z^{1/4}w^{5/4} − 2 z^{5/4}w^{1/4} ) r − A (u) ( z^{7/4}w^{3/4} d A du + 4z^{3/4}w^{7/4} )) Q (x^a) L (x^a) − ( −2 z^{7/4}w^{3/4}A (u) + w^{7/4}z^{3/4}A (u) d A du − 2 z^{5/4}w^{1/4}r d A du ) Q (x^a)

Now to look at the equations that involve D^′. Firstly [3] (6h):

> eqn6h:=D1(Hm)+conj(kappa1)*Hl(u)+tau*Hn+(conj(gamma)-gamma)*Hm-conj(p00):
> P110sbs2(diffsbs(%)):factor(Hlsbs(%));

0 Then we look at [3] (6e):

> eqn6e:=D1(Hn)+(gamma+conj(gamma))*Hn+kappa1*Hm+conj(kappa1)*Hmb:
> P110sbs2(diffsbs(%)):
> factor(Hlsbs(%));

0 and [3] (8c)

> eq8c:=diffsbs(del(p01)+D1(p11)-sigma*p00-2*tau*p01-(rho1+2*gamma)*p11):
> factor(Hlsbs(P110sbs2(%)));

0 and [3] (8b)

> D1(p01)+del(p00)-(tau-2*beta)*p00-2*rho1*p01-kappa1*p11:
> P110sbs2(diffsbs(%)):
> factor(Hlsbs(%));

0 and finally [3] (10d).

> eqn10d:=Psi2*Hm-Psi1*Hn-2*tau*p01-2*gamma*p11+D1(p11):
> P110sbs2(diffsbs(%)):factor(Hlsbs(%));

0 Next we consider what happens if we have a Killing vector (ψ = 0) with ξ_ℓ zero (C = 0)

> subs(C=0,psi=0,Hlsbs(P110sbs2(Hm))); 0 > subs(C=0,psi=0,Hlsbs(P110sbs2(Hn))); 0 Hence we cannot have both ψ and C zero. This is the Collinson and French result:

only one Killing vector at most. We have a look at the homothety.

> Hl(u)*en+Hn*ell-Hm*emb-conj(Hm)*em:
> map(diffsbs,%):
> map(P110sbs2,%):
> map(Hlsbs,%):
> subs(z=x+I*y,w=x-I*y,K):K:=map(expand,%);

K := CA (u) −rC^{d A}_{du} + 2 rψ 2 ψ x − 2 C^{d A}_{du} x 2 ψ y − 2 C^{d A}_{du} y So the obvious Killing vector if A is constant:

> KK:=subs(psi=0,C=1/B,diff(A(u),u)=0,A(u)=B,K):%; 1 0 0 0 The new Killing vector in the other case:

> KK2:=subs(psi=0,A(u)=u*B,C=1/B,K):map(simplify,%); u −r −2x −2y And the proper homothety for both cases: > HH:=subs(C=0,psi=1,K);

0 2r 2x 2y We now turn to the remaining curvature equations and Bianchi identities. To make life easy, we define the weighted derivative operators, [4] section 4.14.

> thorn:=X->diffsbs(diff(X,r)):
> thorn1:=proc(X,p,q)
> local i;
> D1(X)-p*gamma*X-q*conj(gamma)*X;
> diffsbs(expand(%));
> end proc:
> edth:=proc(X,p,q)
> local i;
> del(X)-p*beta*X-q*conj(alpha)*X;
> diffsbs(expand(%))
> end proc:
> edth1:=proc(X,p,q)
> local i;
> del1(X)-p*alpha*X-q*conj(beta)*X;
> diffsbs(expand(%))
> end proc:

As a check on the calculations, we can run through the curvature equations, [4] (4.12.32), some of which we’ve used already, some of which will give us Ψ_3 and Ψ_4. The only ones that do not give zero are (b^′) and (c^′), the first of which gives us Ψ_4:

> thorn1(sigma1,-3,1)-edth1(kappa1,-3,-1)-sigma1*(rho1+conj(rho1))
> +kappa1*(conj(tau)):
> Psi4:=Hlsbs(diffsbs(%));

Ψ4 := −8 L (x^a)^3 z^4Q (x^a)^2 A (u)^2 + (( −12 rz^{7/2}A (u) √ w − 32 A(u)^2z^3w ) Q (x^a)^2 − 8 z^2Q(x^a) ) L (x^a)^2 + [( −32 A (u)^2 z^2w^2 − 8 (A (u))^2 z^3w d A du − 24 rz^{5/2} √ wA (u) ) Q (x^a)^2

+ ( −8 zw − 8 d A du z^2 ) Q (x^a) ] L (x^a) − 8 d A du zw Q (x^a) + ( −12 rA (u) z^{3/2}w^{3/2} − 8 zw^3A (u)^2 − 8A (u)^2 z^2w^2 d du A (u) ) Q(x^a)^2

And (c^′) gives Ψ_3 (as do several others):

> -thorn(kappa1)+conj(tau)*rho1+sigma1*tau:
> Psi3:=diffsbs(%);

Ψ3 := 6 Q (x^a)^2 L (x^a)^2 A (u)^2 z^{13/4}w^{1/4} + (( 14 (A (u))^2 z^{9/4}w^{5/4} + 6

A (u) z^{11/4}r w^{1/4} ) Q (x^a)^2 + 2 Q (x^a) z^{5/4}w^{1/4} ) L (x^a) + ( 2 A (u)^2 z^{9/4}w^{5/4} d A du + 6 z^{5/4}w^{9/4}A (u)^2 + 6 A (u) z^{7/4}w^{3/4}r ) Q (x^a)^2 + 2 d A du z^{5/4}w^{1/4} Q (x^a)

Now we check the leading terms (in inverse powers of r) of our Ψ_3 and Ψ_4 and compare to [1].

> subs(L(u,r,z,w)=Lis,Q(u,r,z,w)=Qis,Hlsbs(Psi4)):
> T4:=subs(r=1/R,%):
> series(T4,R=0,3) assuming R::positive;

−8 d A du zwR^2 + O ( R^3 ) Here the leading term agrees with [1]. Next Ψ_3

> subs(L(u,r,z,w)=Lis,Q(u,r,z,w)=Qis,Hlsbs(Psi3)):
> T3:=subs(r=1/R,%):
> series(T3,R=0,4) assuming R::positive;

2 d A du z^{5/4}w^{1/4}R^2 + 8 z^{7/4}w^{3/4}A (u) R^3 + O ( R^4 ) We find that the leading term agrees with [1], but in the second term the powers of z = ζ and w = ζ are wrong in [1]. We can also check that the Bianchi identities, [4] (4.12.36-39) are satisfied (and they are).

Finally, we turn to the remaining integrability conditions, [3] (10e) to (10h). > P110sbs2(diffsbs(Psi2*Hmb-diff(p00,r)-Psi3*Hl(u))); # [3] 10e 0

> Psi3*Hm-Psi2*Hn-2*rho1*p01+2*beta*p00+del(p00): # [3] 10f
> P110sbs2(diffsbs(%));

0

> Psi3*Hmb-Psi4*Hl(u)+2*sigma1*p01-2*alpha*p00-del1(p00): # [3] 10g
> P110sbs2(diffsbs(%));

0 > Psi4*Hm-Psi3*Hn-2*kappa1*p01+2*gamma*p00+D1(p00): ## [3] 10h

> Hlsbs(P110sbs2(diffsbs(%))); 0 So we see that all the homothetic and Killing equations are satisified and we have shown that there is always a Killing vector in these metrics and also always a homothety.

4 The Calculations (limit cylindrical case)

Since neither [1] not [2] give the spin coefficients for the limit cylindrical metric, we will need to calculate them using Maple’s tensor package. Note that we use the corrected version of this metric, see [5] equation (26.23)

> with(tensor):
> coord:=[u,r,x,y]:g_c:=array(1..4,1..4,symmetric,sparse):
> g_c[1,1]:=-expand(simplify((b+log(r^2*x^4))/x^2/2)
> assuming r::positive,x::positive);

g c_{1,1} := − b 2x^2 − ln (r) x^2 − 2 ln (x) x^2

> g_c[1,2]:=1:g_c[3,3]:=-2*r^2:g_c[4,4]:=-2*x^2:g_c[1,3]:=2*r/x:
> g:=create([-1,-1],eval(g_c)):

Next we calulate all the relevent tensors.

> tensorsGR(coord,g,gup,’detg’, ’C1’,’C2’,’Rm’,’Rc’, ’R’,’G’,’C’); To calculate the spin coefficients, I use a set of routines available on my web site

http://www.maths.unsw.edu.au/∼jds/papers.html
   > read ‘PRcoeff‘:

Now we define the tetrad, with the choice of m^a dictated by the need for the tetrad to be right-handed, so the anti-self duality used in the definition of the homothetic bivector (see [3]) is satisfied.

> md:=create([-1],vector([0,0,r,-I*x])):mup:=raise(gup,md,1):
> mbd:=create([-1],vector([0,0,r,I*x])):mbup:=raise(gup,mbd,1):
> ld:=create([-1],vector([1,0,0,0])):lup:=raise(gup,ld,1):
> nd:=create([-1],vector([g_c[1,1]/2,1,2*r/x,0])):nup:=raise(gup,nd,1):
> his:=linalg[stackmatrix](ld[compts],nd[compts],md[compts],mbd[compts]):
> h:=create([1,-1],op(his)):

Using the routines PRspin and PRcrv from the PRcoeff file we calulate the spin coefficients, and curvature components.

> spins:=PRspin(g,h,C2,coord):
> crv:=PRcurve(g,h,C,Rc,coord);

From these two calculations we find that the non-zero spin coefficients are

τ = β = τ^′ = − 1 2rx , ρ = σ = − 1 2r , γ = − 1 4rx^2 , ρ^′ = σ^′ = b + log(r^2x^4) 8rx^2 ; and the non-zero curvature components are

Ψ_1 = 1 2r^2x

; Ψ_2 = 1 2r^2x^2

; Ψ_3 = b + log(r^2x^4) 8r^2x^3 . Now we define the derivative operators D, D^′, δ and δ^′

> De:=XX->add(lup[compts][i]*diff(XX,coord[i]),i=1..4):
> D1:=XX->add(nup[compts][i]*diff(XX,coord[i]),i=1..4):
> del:=XX->add(mup[compts][i]*diff(XX,coord[i]),i=1..4):
> del1:=XX->add(mbup[compts][i]*diff(XX,coord[i]),i=1..4):

Now to find the Killing vectors. Using [3] (6a),(6c) and (11) gives ξ_ℓ = ξ_ℓ(u). Then from [3] (10a) we get ϕ_{11}, and find that ϕ^0_{11}(u, x, y), the integration constant, is real by [3] (6g), which also gives ξ_m. So

> Hm:=-r*p110(u,x,y)+I*Hm0(u,x,y):Hmb:=-r*p110(u,x,y)-I*Hm0(u,x,y):
> p11:=Hl(u)/2/x/r+p110(u,x,y); # note that p110 is real

p11 :=

Hl (u) 2xr + p110 (u, x, y)

We also solve [3] (10b) for ϕ_{01}.

> crv[Psi1]*Hm-2*spins[sigma]*p01-(spins[beta]-spins[alpha1])*p11+del(p11):
> p01:=expand(solve(%,p01));

p01 := − p110 (u, x, y) 2x −i Hm0 (u, x, y) 2xr − Hl (u) 4x^2r + 1 2 ∂ ∂x p110 (u, x, y)−i r 2x ∂ ∂y p110 (u, x, y) Now looking at [3] (8a), using the fact that κ = 0:

> diff(p01,r)+del1(p11)-2*spins[rho]*p01-
> (spins[tau1]+spins[alpha]-spins[beta1])*p11:
> expand(%);

− 3i 2x ∂ ∂y p110 (u, x, y) So ϕ^0_{11} is independent of y. Now looking at [3] (6b):

> D1(Hl(u))+2*spins[epsilon1]*Hl(u)+spins[tau]*Hm+spins[tau]*Hmb+p01
> +subs(I=-I,p01)-psi:
> collect(%,r);

d du Hl (u) + ∂ ∂x p110 (u, x, y) − ψ So we solve this for ϕ^0_{11}, recalling that ϕ^0_{11} is independent of y, and use it to redefine ϕ_{11} , ϕ_{01} and ξ_m.

> p11:=Hl(u)/2/x/r+(psi-diff(Hl(u),u))*x+p0(u): # note that p0 is real
> p01:=expand(subs(p110(u,x,y)=(psi-diff(Hl(u),u))*x+p0(u),p01)):
> Hm:=expand(subs(p110(u,x,y)=(psi-diff(Hl(u),u))*x+p0(u),Hm));

Hm := −rxψ + rx^d_{du} Hl (u) − rp0 (u) + iHm0 (u, x, y) > Hmb:=expand(subs(p110(u,x,y)=(psi-diff(Hl(u),u))*x+p0(u),Hmb)): Turning to [3] (10c)

> crv[Psi2]*Hl(u)-crv[Psi1]*Hmb-2*spins[rho]*p01-(spins[alpha]-spins[beta1])*p11
> +del1(p11):
> expand(%);

0 Now the right hand side of [3] (6d) is

> -2*spins[epsilon]*Hn-spins[tau1]*Hm-spins[tau1]*Hmb+p01+subs(I=-I,p01)+psi:
> expand(%);

d du Hl (u) − 2 p0 (u) x − Hl (u) 2x^2r This is Dξ_n , so we integrate > int(%,r); r d du Hl (u) − 2 r x p0 (u) −

Hl (u) ln (r) 2x^2

> Hn:=%+Hn0(u,x,y): Turning to [3] (6i), > del(Hm)+spins[sigma1]*Hl(u)+spins[sigma]*Hn+(spins[alpha1]+spins[alpha])*Hm: The coefficients of r are independent, so we collect the terms. > collect(expand(%),r);

−ψ + 1 2 d du Hl (u) − p0 (u) 2x − 1 2x ∂ ∂y Hm0 (u, x, y) − ( 1 2 i ∂ ∂x Hm0 (u, x, y) + Hl (u) b

8x^2 + Hl (u) ln (x) 2x^2 − iHm0 (u, x, y) 2x + 1 2 Hn0 (u, x, y) ) r^{−1} The imaginary part of the r^{−1} term implies ξ^0_m = xf(u, y), for some function f(u, y), so:

> X:=collect(subs(Hm0(u,x,y)=x*f(u,y),%),r):
> Y:=solve(op(5,X),Hn0(u,x,y));

Y := −Hl (u) b + 4 ln (x) 4x^2

> expand(subs(Hn0(u,x,y)=Y,X)); −ψ + 1 2 d du Hl (u) − p0 (u) 2x − 1 2 ∂ ∂y f (u, y) > XX:=rhs(dsolve(%,f(u,y)))*x;

XX := ( −2 yψ + y d du Hl (u) − yp0 (u) x + F1 (u) ) x Where F 1 is an arbitrary function. This XX is ξ^0_m. So

> Hm:=collect(subs(Hm0(u,x,y)=XX,Hm),[r,x,y]);
> Hmb:=collect(subs(Hm0(u,x,y)=XX,Hmb),[r,x,y]):

Hm := [( −ψ + d du Hl (u) ) x − p0 (u) ] r+ [ i ( −2 ψ + d du Hl (u) ) y + i F1 (u) ] x−iy p0 (u)

> Hn:=subs(Hn0(u,x,y)=Y,Hn);

Hn := ( d du Hl (u) ) r − 2 p0 (u) r x − Hl (u) 4x^2 [ 2 ln (r) + b + 4 ln (x) ]

> p01:=expand(subs(Hm0(u,x,y)=XX,p01));

p01 := p0 (u) 2x + iyψ r − iy 2r d du Hl (u) + iy 2rx p0 (u) − i 2r F1 (u) −

Hl (u) 4x^2r Returning to the integrability conditions, we look at [3] (10d)

> eq10d:=crv[Psi2]*Hm-crv[Psi1]*Hn-2*spins[tau]*p01-2*spins[gamma]*p11+D1(p11):
> expand(%);

p0 (u) 2x^2r + d du p0 (u) − x d^2 du^2 Hl (u)

So by comparing coefficients we have > p0(u):=0;Hl:=x->k0*x+k1; p0 (u) := 0 Hl := x → k0 x + k1

And a quick check shows that eqn10d ([3] (10d)) is satisfied. Next, the conjugate of [3] (6h) will give us ϕ_{00}.

> D1(Hmb)+spins[tau]*Hn+(spins[gamma]+spins[epsilon1])*Hmb:
> p00:=collect(expand(%),[psi,k1,k0]);

p00 := ( − b 4x − ln (r) 2x − ln (x) x ) ψ + ( ln (r) 4rx^3 + ln (x) 2rx^3 + b 8rx^3 ) k1 + ( ub 8rx^3 + b 4x − 1 2x + ln (r) 2x + ln (x) x + u ln (x) 2rx^3 + ln (r) u 4rx^3 ) k0 − i ( d du F1 (u) ) x We next check some further integrability conditions, [3] (10e) first.

> crv[Psi3]*Hl(u)-crv[Psi2]*Hmb-2*spins[tau1]*p01+2*spins[epsilon]*p00
> +diff(p00,r):
> expand(%);

0 And then [3] (8d).

> e8d:=diff(p00,r)+del1(p01)-spins[rho]*p00-2*spins[tau1]*p01-spins[sigma1]*p11:
> expand(%);

− ix 2r ( d du F1 (u) )

So the integability function F 1 is constant:

> _F1(u):=k3;expand(e8d); F1 (u) := k3 0 Also [3] (6e) is > eqn6e:=expand(D1(Hn)+2*spins[gamma]*Hn);

eqn6e := − k0 2x^2 Thus k0 = 0, and the coefficients simplify as follows:

> Hl(u);Hn;Hm;

k1 − ln (r) k1 2x^2 − k1 (b + 4 ln (x)) 4x^2 −rxψ + (−2 iyψ + ik3 ) x

> p00;p01;p11; ( − b 4x − ln (r) 2x − ln (x) x ) ψ + ( ln (r) rx^3 + ln (x) 2rx^3 + b 8rx^3 ) k1 iyψ r − ik3 2r − k1 4x^2r k1 2rx + xψ All the remaining homothetic equations and integrability equations are satisfied, and we are left with the general homothetic vector:

> lin_com(Hl(u),nup,Hn,lup,-Hm,mbup,-Hmb,mup);

TABLE ([index char = [1], compts = vector ([k1 , 2 rψ, −xψ, 2 yψ − k3 ])]) That is, k_1∂_u + k_3∂_y + ψ (2r∂_r − x∂_x + 2y∂_y) .

5 Acknowledgments

Maple is a registered trademark of Waterloo Maple Inc.

6 References

[1] Newman and Tamburino 1962 J. Math. Phys. 3 902

[2] Collinson C D and French D C 1967 J. Math. Phys. 8 701

[3] Steele J D 2002 Class. Quantum Grav. 19 259

[4] Penrose R and Rindler W 1984 Spinors and Space-Time vol 1 (Cambridge: Cambridge University Press)

[5] Stephani H, Kramer D, MacCallum M, Hoensaerlars C and Herlt E 2003 Exact Solutions to Einstein’s Field Equations 2nd Edition (Cambridge: Cambridge University Press)

[6] Fayos F and Sopuerta C F 2001 Class. Quantum Grav. 18 353